Journal of Software for Algebra and Geometry

The CotangentSchubert Macaulay2 package

PAUL ZINN-JUSTIN

vol 14 2024

JSAG **14** (2024), 73–85
https://doi.org/10.2140/jsag.2024.14.73

The Journal of Software for
Algebra and Geometry

The CotangentSchubert Macaulay2 package

PAUL ZINN-JUSTIN

ABSTRACT: We present the *Macaulay*2 package *CotangentSchubert*. We show its use in computing motivic Chern and Segre classes of Schubert cells of partial flag varieties and in checking recently found combinatorial formulae for their products.

1. INTRODUCTION. *Cotangent Schubert calculus* is an extension of Schubert calculus, a classical topic in enumerative geometry, that builds upon recent advances in algebraic geometry [3] as well as developments in geometric representation theory and quantum integrable systems [15; 16]; see [17] for a pedagogical introduction.

CotangentSchubert is a *Macaulay2* [13] package which has two main objectives. The first one is to define the basic objects of cotangent Schubert calculus in the context of equivariant K-theory of (partial) flag varieties; that is, motivic Chern and Segre classes of Schubert cells. The second one is to provide an implementation of the formulas of [10; 11; 12], which give the expansion of the product of Segre classes in terms of certain combinatorial gadgets known as *puzzles*. They are a vast generalisation of the original puzzles of Knutson and Tao [9] for ordinary Schubert calculus of Grassmannians, and to differentiate them from the latter, we sometimes call them "generic puzzles". Since these puzzle formulas are rather complicated, a computerised check is most useful.

2. SET-UP AND DEFINITION OF THE RINGS. Let $P\backslash G$ be a (partial) flag variety, where $G = \mathrm{GL}_n(\mathbb{C})$ and P is a parabolic subgroup, with the convention that $B_- \subseteq P$, where B_- is the group of invertible lower triangular matrices. Cotangent Schubert calculus is concerned with cohomology (or K-theory) of $X := T^*(P\backslash G)$, the total space of the cotangent bundle of $P\backslash G$. Nonequivariantly,

$$H^*(X) \cong H^*(P\backslash G)$$

(and similarly in K-theory); however, we always include equivariance with respect to scaling of the fiber of the cotangent bundle, resulting in

$$H^*_{\mathbb{C}^\times}(X) \cong H^*(P\backslash G)[h]$$

(and similarly in K-theory $K_{\mathbb{C}^\times}(X) \cong K(P\backslash G)[t^{\pm}]$), where h (or t) is the equivariant parameter. Furthermore, we may also consider equivariance with respect to the natural action of the Cartan torus $T \subset G$ on X,

MSC2020: 14-04, 14F43.
Keywords: Schubert calculus, motivic Chern classes.
CotangentSchubert version 0.63

leading to rings $H^*_{\hat{T}}(X)$ or $K_{\hat{T}}(X)$ with $\hat{T} = \mathbb{C}^\times \times T$. Finally, in all that follows we consider *localised* cohomology rings in the sense that the base ring (cohomology of a point) is replaced with its fraction field; and in K-theory, we allow ourselves to take the square root of t, defining $q = t^{-1/2}$.

We now turn to the package *CotangentSchubert*. All the code given below assumes that the latter has been loaded with:

```
i1 :  needsPackage "CotangentSchubert";
```

CotangentSchubert offers two different presentations of these rings.

2.1. *The Borel presentation.* In the so-called Borel presentation, the generators of the cohomology / K-theory rings (as algebras over the cohomology of a point) are the Chern classes of tautological bundles (see, e.g., [14, §3.6.4]). In *CotangentSchubert*, the convention is that the generators are Chern classes of *duals* of tautological bundles.

The basic command that creates these rings is `setupCotangent`, whose arguments, specifying the partial flag variety, are the increasing sequence of dimensions of the vector spaces in the filtration:

```
i2 :  (A,B,FF,I)=setupCotangent(2,4,Presentation=>Borel, Ktheory=>false,Equivariant=>false)
```

o2 = $(A,\ B,\ \mathbb{F},\ \{0011,\ 0101,\ 0110,\ 1001,\ 1010,\ 1100\})$

o2 : Sequence

The command returns every ring that is created, allowing the user to name (i.e., globally assign) them. There are three: the first one, A, is the ring $H_{\mathbb{C}^\times}(X)$, where in the example $X = T^* \operatorname{Gr}(2, 4)$; by changing the options `Ktheory` and `Equivariant`, one can obtain instead $K_{\mathbb{C}^\times}(X)$ or equivariance with respect to the whole of $\hat{T}$. The second one, B, is the ring $H_{\mathbb{C}^\times}(T^*(B_-\backslash G))$ of the associated *full* flag variety; according to the splitting principle, A is a subring of B and it is often the most convenient way to consider it. Finally, $\mathbb{F}$ is the base field, i.e., the fraction field of $H_{\mathbb{C}^\times}(\mathrm{pt}) \cong \operatorname{frac}(\mathbb{Z}[h])$.

For convenience, `setupCotangent` also returns the list I of torus fixed points of X in their usual "string" notation, i.e., if $P\backslash G$ is the d-step flag variety

$$\{0 = F_0 < F_1 < \cdots < F_d < F_{d+1} = \mathbb{C}^n\},$$

I is the list of words of n letters in the alphabet $\{0, \ldots, d\}$ such that the number of occurrences of k, $0 \le k \le d$, is $\dim(F_{k+1}/F_k)$. All classes defined below are naturally indexed by such strings.

We can check the presentation of A:

```
i3 :  describe A
```

o3 = $$\frac{\mathbb{F}[x_{1,\{1,2\}}, x_{2,\{1,2\}}, x_{1,\{3,4\}}, x_{2,\{3,4\}}]}{\left(x_{1,\{3,4\}}+x_{1,\{1,2\}},\ -x^2_{1,\{3,4\}}+x_{2,\{3,4\}}+x_{2,\{1,2\}},\ x^3_{1,\{3,4\}}-2\,x_{1,\{3,4\}}x_{2,\{3,4\}},\ x^2_{1,\{3,4\}}x_{2,\{3,4\}}-x^2_{2,\{3,4\}},\ x_{1,\{3,4\}}x^2_{2,\{3,4\}},\ x^3_{2,\{3,4\}}\right)}$$

The notation for variables is that $x_{i,A}$ is the i-th Chern class associated to the subset $A = \{i+1, \ldots, j\}$ corresponding to the tautological bundle F_{k+1}/F_k where $\dim F_k = i$ and $\dim F_{k+1} = j$.

One can check that A has the expected dimension (over $\mathbb{F}$) which is the cardinality of I:

```
i4 :  b=basis A
```

o4 = $\begin{pmatrix} 1 & x_{1,\{1,2\}} & x^2_{1,\{1,2\}} & x_{1,\{1,2\}}x_{2,\{1,2\}} & x_{2,\{1,2\}} & x^2_{2,\{1,2\}} \end{pmatrix}$

o4 : Matrix $A^1 \longleftarrow A^6$

There is a natural embedding from A to B which can be accessed using `promote`:

```
i5 : promote(b,B)
```

o5 = $\begin{pmatrix} 1 & x_2+x_1 & x_2^2+2x_1x_2+x_1^2 & x_1x_2^2+x_1^2x_2 & x_1x_2 & x_1^2x_2^2 \end{pmatrix}$

o5 : Matrix $B^1 \longleftarrow B^6$

where the x_i are the Chern roots; as usual, this can be reversed using `lift`:

```
i6 : lift(oo,A)
```

o6 = $\begin{pmatrix} 1 & x_{1,\{1,2\}} & x_{1,\{1,2\}}^2 & x_{1,\{1,2\}}x_{2,\{1,2\}} & x_{2,\{1,2\}} & x_{2,\{1,2\}}^2 \end{pmatrix}$

o6 : Matrix $A^1 \longleftarrow A^6$

CotangentSchubert implements two types of pushforward to a point. The first one is from X itself, with the method `pushforwardToPointFromCotangent`:

```
i7 : pushforwardToPointFromCotangent b
```

o7 = $\begin{pmatrix} \frac{28}{h^8} & \frac{28}{h^7} & \frac{18}{h^6} & \frac{4}{h^5} & \frac{9}{h^6} & \frac{1}{h^4} \end{pmatrix}$

o7 : Matrix $\mathbb{F}^1 \longleftarrow \mathbb{F}^6$

Since the map $X \to \mathrm{pt}$ is not proper, one obtains denominators.

It is however often useful to pushforward from $P\backslash G$ rather than its cotangent bundle; i.e., classes are first sent to the cohomology ring of $P\backslash G$ by pullback (i.e., restriction to the zero section of X), and then pushed forward to a point:

```
i8 : pushforwardToPoint b
```

o8 = $\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$

o8 : Matrix $\mathbb{F}^1 \longleftarrow \mathbb{F}^6$

Finally, note that one can also access the Chern classes of tautological bundles using the method `tautoClass`:

```
i9 : tautoClass(2,0)
```

o9 = x_1x_2

o9 : B

where the two arguments (i,k) correspond to the i-th Chern class of the vector bundle F_{k+1}/F_k. As with most *CotangentSchubert* functions, an optional argument allows to specify the ring:

```
i10 : tautoClass(2,0,A)
```

o10 = $x_{2,\{1,2\}}$

o10 : A

2.2. ***The equivariant localisation presentation.*** The equivariant localisation theorem (see, e.g., [18]) asserts that after appropriate localisation, the inclusion of the fixed point set $X^{\hat{T}}$ into X induces an isomorphism between $K_{\hat{T}}(X)$ and $K_{\hat{T}}(X^{\hat{T}})$ (and similarly in cohomology); since the torus fixed points of X are isolated, $K_{\hat{T}}(X^{\hat{T}}) \cong \mathbb{F}^{|I|}$ with componentwise product, thus providing a very simple presentation of $K_{\hat{T}}(X)$.

This is implemented in *CotangentSchubert* by changing the option `Presentation`:

```
i11 : (D,FF,I)=setupCotangent(2,4,Presentation=>EquivLoc,Ktheory=>false)
```

o11 = $(D, \mathbb{F}, \{0011, 0101, 0110, 1001, 1010, 1100\})$

o11 : Sequence

Note that this presentation obviously requires `Equivariant=>true` (which is the default). The output is a little different in the sense that the cohomology of the full flag variety is not used in this presentation. Furthermore, A is not encoded as a ring but rather as a vector space with an additional componentwise product:

```
i12 :  tautoClass(2,0)
```

$$\texttt{o12 = } \begin{pmatrix} y_1y_2 \\ y_1y_3 \\ y_1y_4 \\ y_2y_3 \\ y_2y_4 \\ y_3y_4 \end{pmatrix}$$

`o12 :` D

```
i13 :  (tautoClass(1,0))^2*tautoClass(2,1)
```

$$\texttt{o13 = } \begin{pmatrix} (y_2+y_1)^2y_3y_4 \\ y_2(y_3+y_1)^2y_4 \\ y_2y_3(y_4+y_1)^2 \\ y_1(y_3+y_2)^2y_4 \\ y_1y_3(y_4+y_2)^2 \\ y_1y_2(y_4+y_3)^2 \end{pmatrix}$$

`o13 :` D

The y_i are the (cohomological) equivariant parameters of T.

One can go from the Borel presentation (with `Equivariant=>true`) to the equivariant localisation presentation using `restrict`, as will be illustrated in the next section.

Though in what follows, for illustration purposes, we shall mostly use the Borel presentation, it should be noted that the equivariant localisation presentation is more efficient computationally and should be preferred for any flag varieties beyond the smallest ones.

3. Definition of the classes. So far no use has been made of the "cotangent" part of X. We now introduce the main actors of this package, namely motivic classes (which in some ways generalise Schubert classes to the cotangent setting; see also Section 5). There are two main categories of classes

- motivic Chern classes; and
- motivic Segre classes.

The definition of motivic Chern classes can be found in [3]; a very different point of view, closer to cotangent Schubert calculus, can be found in [16] — the equivalence between the two, in an appropriate context, is described in [2; 7]. *CotangentSchubert* computes motivic Chern and Segre classes of Schubert cells inside $P\backslash G$, viewed as classes in $K_{\hat{T}}(X)$ — or, with the option `Ktheory=>false`, Chern and Segre Schwartz–MacPherson classes viewed as classes in $H_{\hat{T}}(X)$ (since the motivic classes are an extension to K-theory of the latter). These classes each come in various flavours to accommodate the differing conventions in the literature. Furthermore, they can be defined with either of the presentations of Section 2.

We shall illustrate these classes with the Borel presentation first:

```
i14 : (A,B,FF,I)=setupCotangent(1,3,Presentation=>Borel,Ktheory=>true,Equivariant=>false)
```

o14 = $(A,\ B,\ \mathbb{F},\ \{011,\ 101,\ 110\})$

o14 : Sequence

Motivic Chern classes are defined using `chernClass`:

```
i15 : chernClass "101"
```

o15 = $\frac{-(2q^2-3)}{q^2}x_2x_3+\frac{q^2-2}{q^2}x_2+\frac{q^2-2}{q^2}x_3+\frac{1}{q^2}$

o15 : B

The argument can be a single string (describing the fixed point contained in the Schubert cell), or a list:

```
i16 : chernClsBorel=chernClass I
```

o16 = $\left(\frac{q^4-6q^2+6}{q^4}x_2x_3+\frac{2q^2-3}{q^4}x_2+\frac{2q^2-3}{q^4}x_3+\frac{1}{q^4}\quad \frac{-(2q^2-3)}{q^2}x_2x_3+\frac{q^2-2}{q^2}x_2+\frac{q^2-2}{q^2}x_3+\frac{1}{q^2}\quad x_2x_3-x_2-x_3+1\right)$

o16 : Matrix $B^1 \longleftarrow B^3$

By default the classes live in B; to obtain them in A, write

```
i17 : lift(chernClsBorel,A)
```

o17 = $\left(\frac{2q^2-3}{q^4}x_{1,\{2,3\}}+\frac{q^4-6q^2+6}{q^4}x_{2,\{2,3\}}+\frac{1}{q^4}\quad \frac{q^2-2}{q^2}x_{1,\{2,3\}}+\frac{-(2q^2-3)}{q^2}x_{2,\{2,3\}}+\frac{1}{q^2}\quad -x_{1,\{2,3\}}+x_{2,\{2,3\}}+1\right)$

o17 : Matrix $A^1 \longleftarrow A^3$

or simply specify the ring explicitly with `chernClass(I,A)` or `chernClass(I,B)`.

In cohomology, `chernClass` produces homogenised versions of Chern–Schwartz–MacPherson classes, where h plays the role of homogeneity parameter; to recover the ordinary (inhomogeneous) classes, one must set $h=-1$.

There is a variant of motivic Chern classes which is accessed via `stableClass`; they correspond to the classes St^λ of [11] (images of fixed points under the stable envelope map with a certain choice of parameters), and we refer to this paper for details; see also [2]. For every type of class, there is also a corresponding *dual* class; in *CotangentSchubert*, the latter have a prime added to their names (e.g., `chernClass'`); in its simplest form, the duality statement is

```
i18 : matrix table(I,I,(i,j)->pushforwardToPointFromCotangent(
                                        stableClass i * stableClass' j))==1
```

o18 = true

Switching to the equivariant setting:

```
i19 : (A,B,FF,I)=setupCotangent(1,3,Presentation=>Borel,Ktheory=>true,Equivariant=>true)
```

o19 = $(A,\ B,\ \mathbb{F},\ \{011,\ 101,\ 110\})$

o19 : Sequence

we can define likewise motivic Segre classes using `segreClass`:

```
i20 : segreClsBorel=segreClass I
```

o20 = $\Bigg(\frac{q^4}{(q^2z_2-z_1)(q^2z_3-z_1)}x_2x_3+\frac{-q^2z_1}{(q^2z_2-z_1)(q^2z_3-z_1)}x_2+\frac{-q^2z_1}{(q^2z_2-z_1)(q^2z_3-z_1)}x_3+\frac{z_1^2}{(q^2z_2-z_1)(q^2z_3-z_1)}$
$\frac{-q^4(q^2z_3+q^2z_2-z_2-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}x_2x_3+\frac{q^2z_2(q^4z_3-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}x_2+\frac{q^2z_2(q^4z_3-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}x_3$
$+\frac{-q^2z_2(q^2z_2z_3+q^2z_1z_3-z_1z_3-z_1z_2)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}$
$\frac{q^4}{(q^2z_3-z_1)(q^2z_3-z_2)}x_2x_3+\frac{-q^4z_3}{(q^2z_3-z_1)(q^2z_3-z_2)}x_2+\frac{-q^4z_3}{(q^2z_3-z_1)(q^2z_3-z_2)}x_3+\frac{q^4z_3^2}{(q^2z_3-z_1)(q^2z_3-z_2)}\Bigg)$

o20 : Matrix $B^1 \longleftarrow B^3$

One proceeds similarly in the equivariant localisation presentation:

```
i21 : (D,FF,I)=setupCotangent(1,3,Presentation=>EquivLoc,Ktheory=>true)
```

o21 = $(D,\ \mathbb{F},\ \{011,\ 101,\ 110\})$

```
o21 : Sequence
```

```
i22 : segreCls=segreClass I
```

$$\texttt{o22} = \begin{pmatrix} 1 & 0 & 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_2-z_1} & \frac{q^2(z_2-z_1)}{q^2z_2-z_1} & 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_3-z_1} & \frac{q^2(q-1)(q+1)z_2(z_3-z_1)}{(q^2z_3-z_1)(q^2z_3-z_2)} & \frac{q^4(z_3-z_1)(z_3-z_2)}{(q^2z_3-z_1)(q^2z_3-z_2)} \end{pmatrix}$$

o22 : Matrix $\mathbb{F}^3 \longleftarrow \mathbb{F}^3$

We can compare the two presentations with `restrict`:

```
i23 : restrict segreClsBorel == segreCls
```

```
o23 = true
```

There is a variant of motivic Segre classes, accessed via `sClass`, which only differs from `segreClass` by a power of q; they correspond to the classes S^λ of [11], and we refer to this paper for details.

In what follows we use exclusively the S classes. We compute their multiplication table as follows:

```
i24 : sCls=sClass I;
```

o24 : Matrix $\mathbb{F}^3 \longleftarrow \mathbb{F}^3$

```
i25 : Table table(I,I,(i,j)->sCls^(-1)*(sClass i * sClass j))
```

$$\texttt{o25} = \begin{matrix} \begin{pmatrix} 1 \\ \frac{-q(q-1)(q+1)z_1}{q^2z_2-z_1} \\ \frac{-q^2(q-1)(q+1)z_1(z_2-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)} \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_2-z_1} \\ \frac{-q(q-1)^2(q+1)^2z_1z_2}{(q^2z_2-z_1)(q^2z_3-z_1)} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_3-z_1} \end{pmatrix} \\ \begin{pmatrix} 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_2-z_1} \\ \frac{-q(q-1)^2(q+1)^2z_1z_2}{(q^2z_2-z_1)(q^2z_3-z_1)} \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{q(z_2-z_1)}{q^2z_2-z_1} \\ \frac{-(q-1)(q+1)z_2(q^4z_2z_3-q^4z_1z_3-q^4z_1z_2+q^2z_1z_2+q^2z_1^2-z_1z_2)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{q(q-1)(q+1)z_2(z_3-z_1)}{(q^2z_3-z_1)(q^2z_3-z_2)} \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_3-z_1} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{q(q-1)(q+1)z_2(z_3-z_1)}{(q^2z_3-z_1)(q^2z_3-z_2)} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{q^2(z_3-z_1)(z_3-z_2)}{(q^2z_3-z_1)(q^2z_3-z_2)} \end{pmatrix} \end{matrix}$$

```
o25 : Expression of class Table
```

4. PUZZLES. In [11], a combinatorial rule to compute the expansion of the product of motivic Segre classes into motivic Segre classes (cf the multiplication table at the end of the previous section) is given in terms of so-called *puzzles*. In *CotangentSchubert* this is implemented for d-step flag varieties, $d \leq 3$ (in principle there is a puzzle rule at $d = 4$ but it is quite complicated, and its implementation is left for future work).

CotangentSchubert allows to draw such puzzles with the method `puzzle`. We now give several examples of use.

4.1. $d = 1$ *puzzles.*

`i26 :  P=puzzle("011","101")`

o26 = $\{$  $\}$

`o26 :  List`

By default, `puzzle` inherits options from the last setup, though it accepts optional arguments to modify this behaviour. In the example, we see the puzzle associated to the product $S^{011}S^{101}$. The input is given by the two strings on the top two sides of the triangle; the output is the string at the bottom of the triangle (all strings are read left-to-right). The puzzle itself is filled with triangles and rhombi (the list of allowed triangle and rhombi labels at $d = 1$ is given in [11, §4.1]). Note that the inside of the puzzle contains labels that do not occur on the boundary (so-called multinumbers; here, 10).

To compute the actual entries of the multiplication table, one associates to each puzzle a *fugacity* (i.e., an element of $\mathbb{F}$ which is the product of fugacities of each elementary triangle or rhombus of the puzzle, also given in [11, §4.1]). This is accessed via `fugacity`:

`i27 :  apply(P,fugacity)`

o27 = $\left\{\frac{(q-1)(q+1)z_1}{q^2z_2-z_1}, \frac{-q\,(q-1)^2(q+1)^2z_1z_2}{(q^2z_2-z_1)(q^2z_3-z_2)}, \frac{q\,(q-1)^2(q+1)^2z_1z_2(z_2-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\right\}$

`o27 :  List`

One must then sum over the puzzles; in *CotangentSchubert* this can be performed with the help of `fugacityVector`:

`i28 :  fugacityVector P`

o28 = $\begin{pmatrix} 0 \\ \frac{(q-1)(q+1)z_1}{q^2z_2-z_1} \\ \frac{-q\,(q-1)^2(q+1)^2z_1z_2}{(q^2z_2-z_1)(q^2z_3-z_1)} \end{pmatrix}$

o28 : $\mathbb{F}^3$

One recognises one of the entries of the multiplication table `o25`.

4.2. $d = 2, 3$ *puzzles.*

One can similarly produce puzzles for 2 or 3-step flag varieties. Let us do an example in the full flag variety of $\mathbb{C}^4$:

`i29 :  (A,FF,I)=setupCotangent(1,2,3,4,Ktheory=>false);`

`i30 :  segreCls=sClass I;`

o30 : Matrix $\mathbb{F}^{24} \longleftarrow \mathbb{F}^{24}$

i31 : λ="3021"; μ="2130";

On the one hand, the product $S^\lambda S^\mu$ can be computed directly:

i33 : lhs=sClass λ $*$ sClass μ;

On the other hand, we can ask for puzzles with sides λ, μ:

```
i34 :  P=puzzle(λ,μ)
```

```
o34 =
```




```
o34 :  List
```

and compute $\sum_\nu c_\nu^{\lambda,\mu} S^\nu$, where $c_\nu^{\lambda,\mu}$ is the sum of fugacities of puzzles with sides λ, μ, ν:

```
i35 :  rhs=segreCls*(fugacityVector P);
```

We check that indeed

```
i36 :  lhs==rhs -- should be true!

o36 =  true
```

4.3. ***Separated descent puzzles.*** Consider two strings λ and μ of the same length n in the alphabets $\_, k, \dots, n-1$ and $0, \dots, k-1, \_$, respectively. For simplicity, we assume here that every letter $0, \dots, n-1$ occurs exactly once in $\lambda \cup \mu$. For example, at $k = 2$, $n = 4$, one could pick

$$\lambda = \_32\_ \quad \text{and} \quad \mu = 1\_0\_.$$

Geometrically, we can think of them as strings indexing fixed points of

$$F\ell(k, k+1, \dots, n) \quad \text{and} \quad F\ell(1, 2, \dots, k, n),$$

respectively, (with nonstandard alphabets $\_ < k < \cdots < n$ and $0 < \cdots < k-1 < \_$). There is a natural map

$$F\ell(1, 2, \dots, n) \xrightarrow{p} F\ell(k, k+1, \dots, n) \times F\ell(1, 2, \dots, k, n),$$

so we can expand

$$p^*(S^\lambda \otimes S^\mu) = \sum_\nu c_\nu^{\lambda,\mu} S^\nu$$

The restriction map for Segre classes is quite simple: it reads

$$p^*(S^\lambda) = \sum_{\mu \prec \lambda} q^{|\mu|-|\lambda|} S^\mu,$$

where the summation is over strings μ that "refine" λ in the natural sense. E.g., starting from the strings $\lambda = \_2\_$ and $\mu = 10\_$, one reindexes them as $\tilde\lambda = 010$ and $\tilde\mu = 102$; then

```
i37 :  (A',B,FF,I')=setupCotangent(2,3,Presentation=>Borel,Ktheory=>true,Equivariant=>true);

i38 :  a=sClass "010"
```

o38 = $\frac{-q\,(q^4z_3-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_3^2+\frac{q\,(q^4z_3^2+q^2z_1z_3+q^2z_1z_2-z_1z_3-z_1z_2-z_1^2)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_3+\frac{-q\,z_1z_3(q^2z_3+q^2z_2-z_2-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}$

o38 : B

```
i39 :  (A,B,FF,I)=setupCotangent(1,2,3,Presentation=>Borel,Ktheory=>true,Equivariant=>true);

i40 :  a==sClass "021" + q * sClass"120" - fill blanks in arbitrary ways

o40 =  true

i41 :  b=sClass "102"
```

$$o41 = \frac{-q^5}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_2x_3^2+\frac{q^3(z_2+z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_2x_3 +\frac{q^3z_2}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_3^2+\frac{-q\,z_1z_2}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_2 +\frac{q\,z_2(q^4z_3-q^2z_3-q^2z_2-z_1)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}\,x_3+\frac{-q\,z_1z_2(q^2z_3-z_3-z_2)}{(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}$$

o41 : B

The corresponding puzzles are described in [12]; they can be obtained with `puzzle`, as we show on an example.

```
i42 :  segreCls=sClass I;
```

o42 : Matrix $B^1 \longleftarrow B^6$

```
i43 :  P=puzzle("_2_","10_",Paths=>true); Table transpose apply(P,p->p,fugacity p)
```

o44 =



$$\frac{(q-1)^2(q+1)^2z_1^2}{(q^2z_2-z_1)(q^2z_3-z_1)}$$



$$\frac{-(q-1)^3(q+1)^3z_1^2z_2}{q\,(q^2z_2-z_1)(q^2z_3-z_1)(q^2z_3-z_2)}$$



$$\frac{(q-1)(q+1)z_2}{q^2z_3-z_2}$$



$$\frac{-q\,(q-1)^2(q+1)^2z_1z_2}{(q^2z_2-z_1)(q^2z_3-z_2)}$$

```
o44 :  Expression of class Table

i45 :  (segreCls*fugacityVector P)_0==a*b

o45 =  true
```

Note the use of the option `Paths=>true` for aesthetic purposes: it shows that the puzzles are simply a collection of (disjoint but possibly intersecting) lattice paths propagating in arbitrary ways with southwest or southeast steps.

4.4. *Puzzles with multinumbers at the bottom.* A similar but distinct product rule can be obtained by taking the pullback of

$$F\ell(j,k,n) \to \mathrm{Gr}(j,n) \times \mathrm{Gr}(k,n),$$

where $j \leq k$, as briefly mentioned in [8, §1.3]. This can be achieved with puzzles where we allow 10s as part of the bottom string; some reindexing is again needed, namely $(0, 10, 1) \mapsto (0, 1, 2)$.

For instance, trying to multiply classes from $\mathrm{Gr}(1, 3)$ and $\mathrm{Gr}(2, 3)$:

```
i46 : (A1,B,FF,I1)=setupCotangent(1,3,Presentation=>Borel,Ktheory=>true,Equivariant=>false);
i47 : a=sClass "101";
i48 : (A2,B,FF,I2)=setupCotangent(2,3,Presentation=>Borel,Ktheory=>true,Equivariant=>false);
i49 : b=sClass "010";
i50 : (A,B,FF,I)=setupCotangent(1,2,3,Presentation=>Borel,Ktheory=>true,Equivariant=>false);
i51 : segreCls=sClass I;
```

`o51 : Matrix` $B^1 \longleftarrow B^6$

```
i52 : a==sClass "102" + q * sClass "201"
o52 = true
i53 : b==sClass "021" + q * sClass "120"
o53 = true
```

is obtained by

```
i54 : P=puzzle("101","010"); Table transpose apply(P,p->p,fugacity p)
```

`o55 =`



$1 \quad 1 \quad \frac{-1}{q} \quad -q$

```
o55 : Expression of class Table
i56 : (segreCls * fugacityVector P)_0==a*b
o56 = true
```

In greater generality, by allowing more multinumbers at the bottom, one can obtain various pullback / product rules; for example, here is one of the puzzles contributing to the product of pullbacks from

$$F\ell(1,3,4,6) \times F\ell(2,3,5,6)$$

to the full flag variety of $\mathbb{C}^6$:

```
i57 :  P=puzzle("312130","013022",Equivariant=>false); P#(random(#P))
```

o58 =




```
o58 :  Puzzle
```

5. Back to ordinary Schubert calculus. Cotangent Schubert calculus can be thought of as a deformation of ordinary Schubert calculus, in the sense that when one sends the equivariant parameter t to zero or infinity (or in cohomology, h to infinity), the leading term of the various motivic classes is simply the corresponding Schubert class (up to various dualities; i.e., one may obtain the class of the structure sheaf of the Schubert variety, or of its dualising sheaf, or the ones obtained from those by the automorphism that sends vector bundles to their duals). For ease of comparison, Schubert classes are also implemented in *CotangentSchubert*:

```
i59 :  (A,B,FF,I)=setupCotangent(1..3,Presentation=>Borel,Ktheory=>false,Equivariant=>false);

i60 :  Gr=schubertClass I
```

o60 = $\begin{pmatrix} 1 & x_2+x_1 & x_1 & x_1^2 & x_1x_2 & x_1^2x_2 \end{pmatrix}$

o60 : Matrix $B^1 \longleftarrow B^6$

For example, in cohomology, one may expand CSM classes into Schubert classes and check positivity of coefficients, in the spirit of [1]:

```
i61 :  CSM=chernClass I;
```

o61 : Matrix $B^1 \longleftarrow B^6$

```
i62 :  sub(inverse basisCoeffs Gr * basisCoeffs CSM, h=>-1)
```

o62 = $\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 1 & 1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 1 & 0 & 0 & 0 \\ 2 & 1 & 2 & 1 & 0 & 0 \\ 2 & 2 & 1 & 0 & 1 & 0 \\ 1 & 1 & 1 & 1 & 1 & 1 \end{pmatrix}$

o62 : Matrix $\mathbb{Q}^6 \longleftarrow \mathbb{Q}^6$

Note the use of `basisCoeffs` to expand an element of a finite-dimensional algebra on the basis of the latter.

One can also produce puzzles for ordinary Schubert calculus, i.e., the classic puzzles of [9] or their extension to higher flag varieties and K-theory in [4; 5; 6; 10], using the option `Generic=>false`:

```
i63 :  P=puzzle("021","102",Generic=>false); Table transpose apply(P,p->p,fugacity p)
```

```
o64 =
```




```
o64 :  Expression of class Table

i65 :  schubertClass "021" * schubertClass "102"==schubertClass "120" + schubertClass "201"

o65 =  true
```

SUPPLEMENT. The online supplement contains version 0.63 of *CotangentSchubert*.

REFERENCES.

[1] P. Aluffi, L. C. Mihalcea, J. Schürmann, and C. Su, “Shadows of characteristic cycles, Verma modules, and positivity of Chern–Schwartz–MacPherson classes of Schubert cells”, preprint, 2017. Zbl arXiv 1709.08697

[2] P. Aluffi, L. C. Mihalcea, J. Schürmann, and C. Su, “Motivic Chern classes of Schubert cells, Hecke algebras, and applications to Casselman’s problem”, preprint, 2019. Zbl arXiv 1902.10101

[3] J.-P. Brasselet, J. Schürmann, and S. Yokura, “Hirzebruch classes and motivic Chern classes for singular spaces”, *J. Topol. Anal.* **2**:1 (2010), 1–55. MR Zbl

[4] A. S. Buch, “A Littlewood–Richardson rule for the K-theory of Grassmannians”, *Acta Math.* **189**:1 (2002), 37–78. MR Zbl

[5] A. S. Buch, “Mutations of puzzles and equivariant cohomology of two-step flag varieties”, *Ann. of Math.* (2) **182**:1 (2015), 173–220. MR Zbl

[6] A. S. Buch, A. Kresch, K. Purbhoo, and H. Tamvakis, “The puzzle conjecture for the cohomology of two-step flag manifolds”, *J. Algebraic Combin.* **44**:4 (2016), 973–1007. MR Zbl

[7] L. M. Fehér, R. Rimányi, and A. Weber, “Motivic Chern classes and K-theoretic stable envelopes”, *Proc. Lond. Math. Soc.* (3) **122**:1 (2021), 153–189. MR Zbl

[8] I. Halacheva, A. Knutson, and P. Zinn-Justin, “Restricting Schubert classes to symplectic Grassmannians using self-dual puzzles”, *Sém. Lothar. Combin.* **82B** (2020), art. id. 83. MR Zbl

[9] A. Knutson and T. Tao, “Puzzles and (equivariant) cohomology of Grassmannians”, *Duke Math. J.* **119**:2 (2003), 221–260. MR Zbl

[10] A. Knutson and P. Zinn-Justin, “Schubert puzzles and integrability, I: invariant trilinear forms”, preprint, 2017. Zbl arXiv 1706.10019

[11] A. Knutson and P. Zinn-Justin, “Schubert puzzles and integrability, II: multiplying motivic Segre classes”, preprint, 2021. Zbl arXiv 2102.00563

[12] A. Knutson and P. Zinn-Justin, “Schubert puzzles and integrability, III: separated descents”, preprint, 2023. Zbl arXiv 2306.13855

[13] D. R. Grayson and M. E. Stillman, “Macaulay2, a software system for research in algebraic geometry”, Available at https://faculty.math.illinois.edu/Macaulay2/. Zbl

[14] L. Manivel, *Symmetric functions, Schubert polynomials and degeneracy loci*, SMF/AMS Texts and Monographs **6**, American Mathematical Society, Providence, RI, 2001. MR Zbl

[15] D. Maulik and A. Okounkov, *Quantum groups and quantum cohomology*, Astérisque **408**, Société Mathématique de France, Paris, 2019. MR Zbl

[16] A. Okounkov, "Lectures on K-theoretic computations in enumerative geometry", pp. 251–380 in *Geometry of moduli spaces and representation theory*, edited by R. Bezrukavnikov et al., IAS/Park City Math. Ser. **24**, American Mathematical Society, Providence, RI, 2017. MR Zbl

[17] R. Rimányi, "Basic notions in cotangent Schubert calculus", lecture notes, 2021, available at https://rimanyi.web.unc.edu/basic-notions-in-cotangent-schubert-calculus/.

[18] I. Rosu, "Equivariant K-theory and equivariant cohomology", *Math. Z.* **243**:3 (2003), 423–448. Zbl

RECEIVED: 20 Dec 2021 REVISED: 21 Jan 2023 ACCEPTED: 6 Feb 2024

PAUL ZINN-JUSTIN:

pzinn@unimelb.edu.au
The University of Melbourne, School of Mathematics and Statistics, Parkville, VIC, Australia